C1M13

Antiderivatives

Fortunately, the topic of *antiderivatives* has nothing to do with social unrest and rebellion within the ranks of mathematicians. It refers to reversing the process of differentiation and determining a new function whose derivative is the function at hand. The social part comes in because when you find one antiderivative you actually get an entire family of functions that serve as antiderivatives of the function. Think about it. If $F'(x) = f(x)$, then $F(x)$ is *an* antiderivative of $f(x)$, and for any constant C, $F(x) + C$ has $f(x)$ as its derivative with respect to x also. Let's formalize this a little.

Definition: A function F is called an *antiderivative* of a function f on an interval I if $F'(x) = f(x)$ for all x in I.

In the plot below you will find a parabola labeled $f(x)$ and a family of cubics with one labeled $F(x)+C1$. Each cubic has $f(x)$ as its derivative, so they are all *antiderivatives of* $f(x)$. We have shown the tangent at $x = 2$ for each and note that the parallel lines all have a slope of -1.



There is a theorem that is usually mentioned in a section on differentiation. Basically, it states that if two functions are defined on an open interval and have the same derivative at all points of that interval, then the two functions differ by a constant. Let's set this up as hypotheses and conclusion.

Theorem:

H1: Functions f and g are defined on an open interval I.

H2: $f'(x) = g'(x)$ for all x in I.

C: There is a constant C for which $f(x) = g(x) + C$ for all x in I.

What does this mean to us here? It means that if we have a collection of functions, all of which are antiderivatives of the same function on an open interval, then each member differs from another by a constant. They are all "shaped the same" and you can get one from the other by translating it up or down. This is true because $f'(x) = g'(x) \Rightarrow (f' - g')(x) = f'(x) - g'(x) = 0$ and only constant functions have zero for their derivative. This last statement is yet another consequence of the **MVT**.

Maple Example: Use Maple to find an antiderivative $F(x)$ of $f(x) = e^{2x}$ that satisfies $F(1.2) = 4$. The condition $F(1.2) = 4$ is called a *boundary value* for F.

The terms *antiderivative* and *integral* are commonly used interchangeably even though that usage is imprecise. Maple uses `int` as a command to find either for a function. The use of a lowercase "i" makes the command active, while an uppercase "I" is inactive and some form of evaluation is necessary to activate it.

```
> restart:    with(plots):    with(plottools):
> f:=x->exp(2*x);
```

$$f := x \rightarrow \mathbf{e}^{(2x)}$$

Now we will find an antiderivative for $f(x)$.

```
> A:=int(f(x),x);
```

$$A := \frac{1}{2}\mathbf{e}^{(2x)}$$

```
> F:=unapply(A+C,x);
```

$$F := x \to \frac{1}{2}\mathbf{e}^{(2x)} + C$$

This is how we add a constant to our expression and then make a function out of the result. Next we force $F(1.2) = 4$ and solve for the constant C that satisfies this condition.

```
> C:=solve(F(1.2)=4,C);
```

$$C := -1.511588190$$

```
> F(x);
```

$$\frac{1}{2}\mathbf{e}^{(2x)} - 1.511588190$$

```
> evalf(F(1.2));
```

$$4.000000000$$

```
> diff(F(x),x);
```

$$\mathbf{e}^{(2x)}$$

The last two lines are simply to check that we got the results that we wanted. You should be able to produce the plot which follows.



Maple Example: A stone is thrown upwards from a cliff with a velocity of 48 feet per second and it lands 1120 feet below. Using the point where it lands as the reference point, determine when the stone lands and how fast it is going. Gravitational acceleration is 32 feet per second per second, *downwards*. This means that we are given *two* boundary values. If $a(t)$ denotes acceleration and $v(t)$ denotes velocity, then

$$a(0) = -32 \qquad\qquad v(0) = 48$$

```
> restart:
> a:=-32;
```

$$a := -32$$

```
> v:=unapply(int(a,t)+C1,t);
```

$$v := t \to -32\,t + C1$$

Antidifferentiate $a(t)$, add a constant, make v into a function of t.

```
> C1:=solve(v(0)=48,C1);
```

$$C1 := 48$$

```
> v(t);
```

$$-32\,t + 48$$

```
> p:=unapply(int(v(t),t)+C2,t);
```

$$p := t \to -16\,t^2 + 48\,t + C2$$

Antidifferentiate $v(t)$, add a constant, make p into a function of t.

```
> C2:=solve(p(0)=1120,C2);
```

$$C2 := 1120$$

```
> p(t);
```

$$-16\,t^2 + 48\,t + 1120$$

```
> t1:=solve(p(t)=0,t);
```

$$t1 := -7, 10$$

```
> t1:=t1[2];
```

$$t1 := 10$$

```
> v(t1);
```

$$-272$$

We see that the first solution to $p(t) = 0$ is negative, which is impossible, so we select the second solution, which is positive. The stone hits the ground 10 seconds after it is tossed upwards, striking the ground at 272 feet per second *downwards*.

C1M13 Problems: Use Maple to solve the problems and plot the graphs.

1. A stone is thrown *downwards* from a cliff with a velocity of 48 feet per second and it lands 1120 feet below. Using the point where it lands as the reference point, determine when the stone lands and how fast it is going.

2. Find an antiderivative F of $f(x) = \arctan(x)$ that satisfies $F(1) = \pi$.

3. Suppose that $g(x) = \sin(x) + e^{2x}$, G is an antiderivative of g, and H is an antiderivative of G. If $G(0) = 3$ and $H(0) = 5$, find $H(x)$ and $H(\pi)$.